
1 Grape

This manual describes the GRAPE (Version 4.1) package for computing with graphs and groups.

GRAPE is primarily designed for the construction and analysis of finite graphs related to groups, designs or
geometries. Special emphasis is placed on the determination of regularity properties and subgraph structure.
The GRAPE philosophy is that a graph gamma always comes together with a known subgroup G of the
automorphism group of gamma, and that G is used to reduce the storage and CPU-time requirements
for calculations with gamma (see [Soi93]). Of course G may be the trivial group, and in this case GRAPE
algorithms may perform more slowly than strictly combinatorial algorithms (although this degradation in
performance is hopefully never more than a fixed constant factor).

Most GRAPE functions are written entirely in the GAP language. However, the GRAPE functions Automor-
phismGroup, AutGroupGraph, IsIsomorphicGraph and PartialLinearSpaces make direct or indirect use
of B. D. McKay’s nauty (Version 2.0b5) package [McK90], via a GRAPE interface. These functions can only
be used on a fully installed version of GRAPE in a UNIX environment. Installation of GRAPE is described
in its README file and in its manual section 1.1.

Except for the nauty package included with GRAPE, the GRAPE package was designed and written by
Leonard H. Soicher, School of Mathematical Sciences, Queen Mary, University of London.

If you use GRAPE to solve a problem then please send a short email about it to L.H.Soicher@qmul.ac.uk,
and reference the GRAPE package as follows:

Leonard H. Soicher. GRAPE: a system for computing with graphs and groups. In Larry Finkelstein and
William M. Kantor, editors, Groups and Computation, volume 11 of DIMACS Series in Discrete Mathe-
matics and Theoretical Computer Science, pages 287–291. American Mathematical Society, 1993. GRAPE
homepage:

http://www.gap-system.org/Share/grape.html .

If your work made use a function depending on the nauty package then you should also reference nauty as
follows:

Brendan D. McKay. nauty user’s guide (version 1.5), Technical report TR-CS-90-02. Australian National
University, Computer Science Department, 1990. nauty homepage:

http://cs.anu.edu.au/people/bdm/nauty/ .

The development of GRAPE was partially supported by a European Union HCM grant in “Computational
Group Theory”.

1.1 Installing the GRAPE Package

To install GRAPE 4.1 (after installing GAP), first obtain the GRAPE archive file grape4r1.zoo, available
from

http://www.gap-system.org/Share/grape.html and then copy this archive file into the pkg di-
rectory of the GAP root directory. Actually, it is possible to have several GAP root directories (see 9.2), and
so it is easy to install GRAPE locally even if you have no permission to add files to the main GAP installation
(see 74.2). Now go to the appropriate pkg directory containing grape4r1.zoo, and then run

unzoo -x grape4r1.zoo

to unpack GRAPE.

After unpacking GRAPE, the GAP-only part of GRAPE is installed. The parts of GRAPE depending on
B. D. Mckay’s nauty package (for computing automorphism groups and testing graph isomorphism) are only
available in a UNIX environment, where you should proceed as follows:

Go to the newly created grape directory (which we shall refer to as the home directory of GRAPE) and
call /bin/sh configure path, where path is the path to the home directory of the GAP distribution. So for
example, if you install GRAPE in the pkg directory of the GAP home directory, call

/bin/sh configure ../..

This will fetch the name of the system architecture on which GAP has been compiled, and create a Makefile.
Now call

make

to give a list of target architectures for which nauty can be compiled within GRAPE. Select the most suitable
target and type

make <target>

to create the binaries and to put them in the appropriate place. For example, if you are on a LINUX system
with gcc, type

make linux-gcc

This completes the installation of GRAPE for a single architecture. If GAP is installed on different architec-
tures on a common file system, this configuration process will only work for the last architecture for which
GAP was compiled. Therefore, you should always follow the above procedure to install the GRAPE binaries
immediately after compiling GAP on a given architecture. However, if you want to add GRAPE later, you
can just call configure again in the main GAP directory for the architecture before installing the GRAPE
binaries for that architecture.

You should now test GRAPE and the interface to nauty on each architecture on which you have installed
GRAPE. Start up GAP and at the prompt type

RequirePackage( "grape" );

On-line documentation for GRAPE should be available by typing

?GRAPE

The command

IsIsomorphicGraph( JohnsonGraph(7,3), JohnsonGraph(7,4) );

should return true, and

Size( AutGroupGraph( JohnsonGraph(4,2) ) );

should be 48.

Both dvi and PostScript versions of the GRAPE manual are available (as manual.dvi and manual.ps re-
spectively) in the doc directory of the home directory of GRAPE.

If you install GRAPE, then please tell L.H.Soicher@qmul.ac.uk, where you should also send any
comments or bug reports.

1.2 Loading GRAPE

Before using GRAPE you must load the package within GAP by calling the statement

gap> RequirePackage("grape");

Loading GRAPE 4.1 (GRaph Algorithms using PErmutation groups),
by L.H.Soicher@qmul.ac.uk.

true

1.3 The structure of a graph in GRAPE

In general GRAPE deals with finite directed graphs which may have loops but have no multiple edges.
However, many GRAPE functions only work for simple graphs (i.e. no loops, and whenever [x , y ] is an edge
then so is [y , x ]), but these functions will check if an input graph is simple.

In GRAPE, a graph gamma is stored as a record, with mandatory components isGraph, order, group,
schreierVector, representatives, and adjacencies. Usually, the user need not be aware of this record
structure, and is strongly advised only to use GRAPE functions to construct and modify graphs.

The order component contains the number of vertices of gamma. The vertices of gamma are always
1,2,...,gamma.order, but they may also be given names, either by a user (using AssignVertexNames)
or by a function constructing a graph (e.g. InducedSubgraph, BipartiteDouble, QuotientGraph). The
names component, if present, records these names, with gamma.names[i] the name of vertex i . If the
names component is not present (the user may, for example, choose to unbind it), then the names are
taken to be 1,2,...,gamma.order. The group component records the GAP permutation group associated
with gamma (this group must be a subgroup of the automorphism group of gamma). The representa-
tives component records a set of orbit representatives for the action of gamma.group on the vertices of
gamma, with gamma.adjacencies[i] being the set of vertices adjacent to gamma.representatives[i].
The group and schreierVector components are used to compute the adjacency-set of an arbitrary vertex
of gamma (this is done by the function Adjacency).

The only mandatory component which may change once a graph is initially constructed is adjacencies
(when an edge-orbit of gamma.group is added to, or removed from, gamma). A graph record may also have
some of the optional components isSimple, autGroup, and canonicalLabelling, which record information
about that graph.

Note All global variables used by GRAPE start with GRAPE .

1.4 Examples of the use of GRAPE

We give here a simple example to illustrate the use of GRAPE. All functions used are described in detail in
this manual. More sophisticated examples of the use of GRAPE can be found in chapter 9, as part of the the
web-page

http://www.maths.qmul.ac.uk/~leonard/partialspreads/ , and also in the references [Cam99],
[CSS99] and [HL99].

In the example here, we construct the Petersen graph P , and its edge graph (also called line graph) EP . We
compute the global parameters of EP , and so verify that EP is distance-regular (see [BCN89]).

gap> RequirePackage("grape");

Loading GRAPE 4.1 (GRaph Algorithms using PErmutation groups),
by L.H.Soicher@qmul.ac.uk.

true
gap> P := Graph( SymmetricGroup(5), [[1,2]], OnSets,
> function(x,y) return Intersection(x,y)=[]; end );
rec( isGraph := true, order := 10,
group := Group([ ( 1, 2, 3, 5, 7)( 4, 6, 8, 9,10), ( 2, 4)( 6, 9)( 7,10) ]),
schreierVector := [ -1, 1, 1, 2, 1, 1, 1, 1, 2, 2 ],
adjacencies := [ [ 3, 5, 8 ] ], representatives := [ 1 ],
names := [ [ 1, 2 ], [ 2, 3 ], [ 3, 4 ], [ 1, 3 ], [ 4, 5 ], [ 2, 4 ],

[ 1, 5 ], [ 3, 5 ], [ 1, 4 ], [ 2, 5 ] ] )
gap> Diameter(P);
2
gap> Girth(P);
5
gap> EP := EdgeGraph(P);
rec( isGraph := true, order := 15,
group := Group([ ( 1, 4, 7, 2, 5)( 3, 6, 8, 9,12)(10,13,14,15,11),

( 4, 9)( 5,11)( 6,10)( 7, 8)(12,15)(13,14) ]),
schreierVector := [ -1, 1, 1, 1, 1, 1, 1, 2, 2, 1, 2, 1, 1, 1, 2 ],
adjacencies := [ [ 2, 3, 7, 8 ] ], representatives := [ 1 ],
isSimple := true,
names := [ [ [ 1, 2 ], [ 3, 4 ] ], [ [ 1, 2 ], [ 4, 5 ] ],

[ [ 1, 2 ], [ 3, 5 ] ], [ [ 2, 3 ], [ 4, 5 ] ], [ [ 2, 3 ], [ 1, 5 ] ],
[ [ 2, 3 ], [ 1, 4 ] ], [ [ 3, 4 ], [ 1, 5 ] ], [ [ 3, 4 ], [ 2, 5 ] ],
[ [ 1, 3 ], [ 4, 5 ] ], [ [ 1, 3 ], [ 2, 4 ] ], [ [ 1, 3 ], [ 2, 5 ] ],
[ [ 2, 4 ], [ 1, 5 ] ], [ [ 2, 4 ], [ 3, 5 ] ], [ [ 3, 5 ], [ 1, 4 ] ],
[ [ 1, 4 ], [ 2, 5 ] ] ] )

gap> GlobalParameters(EP);
[ [ 0, 0, 4 ], [ 1, 1, 2 ], [ 1, 2, 1 ], [ 4, 0, 0 ] ]

Functions to construct
and modify graphs

This chapter describes the functions used to construct and modify graphs.

2.1 Graph
1I Graph( G, L, act, rel )
I Graph( G, L, act, rel, invt )

This is the most general and useful way of constructing a graph in GRAPE.
First suppose that the optional boolean parameter invt is unbound or has value false. Then L should be
a list of elements of a set S on which the group G acts, with the action given by the function act . The
parameter rel should be a boolean function defining a G-invariant relation on S (so that for g in G , x , y in
S , rel(x , y) if and only if rel(act(x , g), act(y , g))). Then the function Graph returns a graph gamma which
has as vertex-names (an immutable copy of)

Concatenation( Orbits( G, L, act ) )

(the concatenation of the distinct orbits of the elements in L under G), and for vertices v ,w of gamma,
[v ,w ] is an edge if and only if

rel( VertexName( gamma, v ), VertexName( gamma, w ) ).
Now if the parameter invt exists and has value true, then it is assumed that L is invariant under G with
respect to action act . Then the function Graph behaves as above, except that the vertex-names of gamma
become (an immutable copy of) L.
The group associated with the graph gamma returned is the image of G acting via act on gamma.names.
For example, suppose you have an n by n adjacency matrix A for a graph X , so that the vertex-set of
X is {1, . . . ,n}, and [i , j ] is an edge of X if and only if A[i ][j ] = 1. Suppose also that G ≤ Aut (X ) (G
may be trivial). Then you can make a GRAPE graph isomorphic to X via Graph( G, [1..n], OnPoints,
function(x,y) return A[x][y]=1; end, true );

gap> A := [[0,1,0],[1,0,0],[0,0,1]];
[ [ 0, 1, 0 ], [ 1, 0, 0 ], [ 0, 0, 1 ] ]
gap> G := Group( (1,2) );
Group([ (1,2) ])
gap> Graph( G, [1..3], OnPoints,
> function(x,y) return A[x][y]=1; end,
> true );
rec(

isGraph := true,
order := 3,
group := Group( [ (1,2) ] ),
schreierVector := [ -1, 1, -2 ],
adjacencies := [ [ 2 ], [ 3 ] ],
representatives := [ 1, 3 ],
names := [ 1, 2, 3 ] )

We now use Graph to construct the Petersen graph.

gap> Petersen := Graph( SymmetricGroup(5), [[1,2]], OnSets,
> function(x,y) return Intersection(x,y)=[]; end );
rec(
isGraph := true,
order := 10,
group := Group( [ ( 1, 2, 3, 5, 7)( 4, 6, 8, 9,10), ( 2, 4)( 6, 9)( 7,10)

] ),
schreierVector := [ -1, 1, 1, 2, 1, 1, 1, 1, 2, 2 ],
adjacencies := [ [ 3, 5, 8 ] ],
representatives := [ 1 ],
names := [ [ 1, 2 ], [ 2, 3 ], [ 3, 4 ], [ 1, 3 ], [ 4, 5 ], [ 2, 4 ],

[ 1, 5 ], [ 3, 5 ], [ 1, 4 ], [ 2, 5 ] ] )

2.2 EdgeOrbitsGraph
1I EdgeOrbitsGraph( G, edges )
I EdgeOrbitsGraph( G, edges, n )

This is a common way of constructing a graph in GRAPE.
This function returns the (directed) graph with vertex-set {1, . . . ,n}, edge-set ∪e∈edges eG , and associated
(permutation) group G , which must act naturally on {1, . . . ,n}. The parameter edges should be a list of
edges (lists of length 2 of vertices), although a singleton edge will be understood as an edge-list of length 1.
The parameter n may be omitted, in which case n is taken to be the largest point moved by G .
Note that G may be the trivial permutation group (Group( () ) in GAP notation), in which case the
(directed) edges of gamma are simply those in the list edges.

gap> EdgeOrbitsGraph( Group((1,3),(1,2)(3,4)), [[1,2],[4,5]], 5 );
rec(
isGraph := true,
order := 5,
group := Group( [ (1,3), (1,2)(3,4) ] ),
schreierVector := [ -1, 2, 1, 2, -2 ],
adjacencies := [ [ 2, 4, 5 ], [ ] ],
representatives := [ 1, 5 ],
isSimple := false )

2.3 NullGraph
1I NullGraph( G )
I NullGraph( G, n )

This function returns the null graph (graph with no edges) with vertex-set {1, . . . ,n}, and associated (per-
mutation) group G . The parameter n may be omitted, in which case n is taken to be the largest point
moved by G .
See also 3.20.1.

gap> NullGraph( Group( (1,2,3) ), 4 );
rec(
isGraph := true,
order := 4,
group := Group( [ (1,2,3) ] ),
schreierVector := [ -1, 1, 1, -2 ],
adjacencies := [ [ ], [ ] ],
representatives := [ 1, 4 ],
isSimple := true )

2.4 CompleteGraph

1I CompleteGraph( G )
I CompleteGraph( G, n )
I CompleteGraph( G, n, mustloops )

This function returns the complete graph with vertex-set {1, . . . ,n} and associated (permutation) group
G . The parameter n may be omitted, in which case n is taken to be the largest point moved by G . The
optional boolean parameter mustloops determines whether the complete graph has all loops present or no
loops (default: false (no loops)).

See also 3.21.1.

gap> CompleteGraph( Group( (1,2,3), (1,2) ) );
rec(
isGraph := true,
order := 3,
group := Group( [ (1,2,3), (1,2) ] ),
schreierVector := [ -1, 1, 1 ],
adjacencies := [ [ 2, 3 ] ],
representatives := [ 1 ],
isSimple := true )

2.5 JohnsonGraph

1I JohnsonGraph( n, e )

Let n and e be integers, with n ≥ e ≥ 0. Then this function returns a graph gamma isomorphic to the
Johnson graph J (n, e). The vertices (actually the vertex-names) of gamma are the e-subsets of {1, . . . ,n},
with x joined to y if and only if |x ∩ y | = e − 1. The group associated with gamma is the image of the
symmetric group Sn acting on the e-subsets of {1, . . . ,n}.

gap> JohnsonGraph(5,3);
rec(
isGraph := true,
order := 10,
group := Group( [ ( 1, 7,10, 6, 3)( 2, 8, 4, 9, 5), ( 4, 7)( 5, 8)( 6, 9)

] ),
schreierVector := [ -1, 1, 1, 2, 1, 1, 1, 2, 1, 1 ],
adjacencies := [ [ 2, 3, 4, 5, 7, 8 ] ],
representatives := [ 1 ],
names := [ [ 1, 2, 3 ], [ 1, 2, 4 ], [ 1, 2, 5 ], [ 1, 3, 4 ], [ 1, 3, 5 ],

[ 1, 4, 5 ], [ 2, 3, 4 ], [ 2, 3, 5 ], [ 2, 4, 5 ], [ 3, 4, 5 ] ],
isSimple := true )

2.6 CayleyGraph

1I CayleyGraph( G )
I CayleyGraph( G, gens )
I CayleyGraph( G, gens, undirected )

Given a group G and a generating list gens for G , CayleyGraph( G, gens ) returns a Cayley graph for
G with respect to gens. The generating list gens is optional, and if omitted, then gens is taken to be
GeneratorsOfGroup( G ). The boolean argument undirected is also optional, and if undirected=true (the
default), then the returned graph is undirected (as if gens was closed under inversion, whether or not it is).



12 Chapter 2. Functions to construct and modify graphs

The Cayley graph caygraph which is returned is defined as follows: the vertices (actually the vertex-names)
of caygraph are the elements of G ; if undirected=true (the default) then vertices x , y are joined by an edge
if and only if there is a g in the list gens with y = gx or y = g−1x ; if undirected=false then [x , y ] is an
edge if and only if there is a g in gens with y = gx .

The permutation group caygraph.group associated with caygraph is the image of G acting in its right regular
representation.

Note It is not checked whether G is actually generated by gens. However, even if G is not generated by
gens, the function still works as described above (as long as gens is contained in G), but returns a “Cayley
graph” which is not connected.

gap> C:=CayleyGraph(SymmetricGroup(4),[(1,2),(2,3),(3,4)]);
rec(
isGraph := true,
order := 24,
group :=
Group( [ ( 1,10,17,19)( 2, 9,18,20)( 3,12,14,21)( 4,11,13,22)( 5, 7,16,23)

( 6, 8,15,24), ( 1, 7)( 2, 8)( 3, 9)( 4,10)( 5,11)( 6,12)(13,15)
(14,16)(17,18)(19,21)(20,22)(23,24) ] ),

schreierVector := [ -1, 1, 1, 2, 2, 1, 2, 2, 2, 1, 1, 1, 1, 2, 2, 1, 1, 2,
1, 1, 2, 2, 1, 2 ],

adjacencies := [ [ 2, 3, 7 ] ],
representatives := [ 1 ],
names := [ (), (3,4), (2,3), (2,3,4), (2,4,3), (2,4), (1,2), (1,2)(3,4),

(1,2,3), (1,2,3,4), (1,2,4,3), (1,2,4), (1,3,2), (1,3,4,2), (1,3),
(1,3,4), (1,3)(2,4), (1,3,2,4), (1,4,3,2), (1,4,2), (1,4,3), (1,4),
(1,4,2,3), (1,4)(2,3) ],

isSimple := true )
gap> Girth(C);
4
gap> Diameter(C);
6

2.7 AddEdgeOrbit

1I AddEdgeOrbit( gamma, e )
I AddEdgeOrbit( gamma, e, H )

This procedure adds the orbit of e under gamma.group to the edge-set of the graph gamma. The parameter
e must be a sequence of length 2 of vertices of gamma. If the optional third parameter H is given then it
is assumed that e[2] has the same orbit under H as it does under the stabilizer in gamma.group of e[1],
and this knowledge can speed up the procedure.

Note that if gamma.group is trivial then this procedure simply adds the single (directed) edge e to gamma.

See also 2.8.1.

gap> gamma := NullGraph( Group( (1,3), (1,2)(3,4) ) );;
gap> AddEdgeOrbit( gamma, [4,3] );
gap> gamma;
rec(
isGraph := true,
order := 4,
group := Group( [ (1,3), (1,2)(3,4) ] ),
schreierVector := [ -1, 2, 1, 2 ],

adjacencies := [ [ 2, 4 ] ],
representatives := [ 1 ],
isSimple := true )

gap> GlobalParameters(gamma);
[ [ 0, 0, 2 ], [ 1, 0, 1 ], [ 2, 0, 0 ] ]

2.8 RemoveEdgeOrbit

1I RemoveEdgeOrbit( gamma, e )
I RemoveEdgeOrbit( gamma, e, H )

This procedure removes the orbit of e under gamma.group from the edge-set of the graph gamma. The
parameter e must be a sequence of length 2 of vertices of gamma, but if e is not an edge of gamma then
this procedure has no effect. If the optional third parameter H is given then it is assumed that e[2] has the
same orbit under H as it does under the stabilizer in gamma.group of e[1], and this knowledge can speed
up the procedure.

See also 2.7.1.

gap> gamma := CompleteGraph( Group( (1,3), (1,2)(3,4) ) );;
gap> RemoveEdgeOrbit( gamma, [1,3] );
gap> gamma;
rec(
isGraph := true,
order := 4,
group := Group( [ (1,3), (1,2)(3,4) ] ),
schreierVector := [ -1, 2, 1, 2 ],
adjacencies := [ [ 2, 4 ] ],
representatives := [ 1 ],
isSimple := true )

gap> GlobalParameters(gamma);
[ [ 0, 0, 2 ], [ 1, 0, 1 ], [ 2, 0, 0 ] ]

2.9 AssignVertexNames

1I AssignVertexNames( gamma, names )

This procedure allows the user to give new names for the vertices of gamma, by specifying a list names (of
length gamma.order) of vertex-names for the vertices of gamma, such that names[i] contains the user’s
name for the i -th vertex of gamma.

An immutable copy of names is assigned to gamma.names.

See also 3.5.1 and 3.4.1.

gap> gamma := NullGraph( Group(()), 3 );
rec(
isGraph := true,
order := 3,
group := Group( [ () ] ),
schreierVector := [ -1, -2, -3 ],
adjacencies := [ [ ], [ ], [ ] ],
representatives := [ 1, 2, 3 ],
isSimple := true )

gap> AssignVertexNames( gamma, ["a","b","c"] );
gap> gamma;



14 Chapter 2. Functions to construct and modify graphs

rec(
isGraph := true,
order := 3,
group := Group( [ () ] ),
schreierVector := [ -1, -2, -3 ],
adjacencies := [ [ ], [ ], [ ] ],
representatives := [ 1, 2, 3 ],
isSimple := true,
names := [ "a", "b", "c" ] )



3

Functions to
inspect graphs,

vertices and edges

This chapter describes functions to inspect graphs, vertices and edges.

3.1 IsGraph

1I IsGraph( obj )

This boolean function returns true if and only if obj , which can be an object of arbitrary type, is a graph.

gap> IsGraph( 1 );
false
gap> IsGraph( JohnsonGraph( 3, 2 ) );
true

3.2 OrderGraph

1I OrderGraph( gamma )

This function returns the number of vertices (the order) of the graph gamma.

gap> OrderGraph( JohnsonGraph( 4, 2 ) );
6

3.3 IsVertex

1I IsVertex( gamma, v )

This boolean function returns true if and only if v is vertex of the graph gamma.

gap> gamma := JohnsonGraph( 3, 2 );;
gap> IsVertex( gamma, 1 );
true
gap> IsVertex( gamma, 4 );
false

3.4 VertexName

1I VertexName( gamma, v )

This function returns (an immutable copy of) the name of vertex v in the graph gamma.

See also 3.5.1 and 2.9.1.

gap> VertexName( JohnsonGraph(4,2), 6 );
[ 3, 4 ]

3.5 VertexNames

1I VertexNames( gamma )

This function returns (an immutable copy of) the list of vertex-names for the graph gamma. The i -th element
of this list is the name of vertex i .

See also 3.4.1 and 2.9.1.

gap> VertexNames( JohnsonGraph(4,2) );
[ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 2, 3 ], [ 2, 4 ], [ 3, 4 ] ]

3.6 Vertices

1I Vertices( gamma )

This function returns the vertex-set {1, . . . , gamma.order} of the graph gamma.

gap> Vertices( JohnsonGraph( 4, 2 ) );
[ 1 .. 6 ]

3.7 VertexDegree

1I VertexDegree( gamma, v )

This function returns the (out)degree of the vertex v of the graph gamma.

gap> VertexDegree( JohnsonGraph( 3, 2 ), 1 );
2

3.8 VertexDegrees

1I VertexDegrees( gamma )

This function returns the set of vertex (out)degrees for the graph gamma.

gap> VertexDegrees( JohnsonGraph( 4, 2 ) );
[ 4 ]

3.9 IsLoopy

1I IsLoopy( gamma )

This boolean function returns true if and only if the graph gamma has a loop, i.e. an edge of the form
[x , x ].

gap> IsLoopy( JohnsonGraph( 4, 2 ) );
false
gap> IsLoopy( CompleteGraph( Group( (1,2,3), (1,2) ), 3 ) );
false
gap> IsLoopy( CompleteGraph( Group( (1,2,3), (1,2) ), 3, true ) );
true

3.10 IsSimpleGraph

1I IsSimpleGraph( gamma )

This boolean function returns true if and only if the graph gamma is simple, i.e. has no loops and whenever
[x , y ] is an edge then so is [y , x ].

gap> IsSimpleGraph( CompleteGraph( Group( (1,2,3) ), 3 ) );
true
gap> IsSimpleGraph( CompleteGraph( Group( (1,2,3) ), 3, true ) );
false

3.11 Adjacency

1I Adjacency( gamma, v )

This function returns (a copy of) the set of vertices of the graph gamma adjacent to the vertex v of gamma.
A vertex w is adjacent to v if and only if [v ,w ] is an edge.

gap> Adjacency( JohnsonGraph( 4, 2 ), 1 );
[ 2, 3, 4, 5 ]
gap> Adjacency( JohnsonGraph( 4, 2 ), 6 );
[ 2, 3, 4, 5 ]

3.12 IsEdge

1I IsEdge( gamma, e )

This boolean function returns true if and only if e is an edge of the graph gamma.

gap> IsEdge( JohnsonGraph( 4, 2 ), [ 1, 2 ] );
true
gap> IsEdge( JohnsonGraph( 4, 2 ), [ 1, 6 ] );
false

3.13 DirectedEdges

1I DirectedEdges( gamma )

This function returns the set of directed (ordered) edges of the graph gamma.

See also 3.14.1.

gap> gamma := JohnsonGraph( 4, 3 );
rec( isGraph := true, order := 4, group := Group([ (1,4,3,2), (3,4) ]),
schreierVector := [ -1, 1, 1, 1 ], adjacencies := [ [ 2, 3, 4 ] ],
representatives := [ 1 ],
names := [ [ 1, 2, 3 ], [ 1, 2, 4 ], [ 1, 3, 4 ], [ 2, 3, 4 ] ],
isSimple := true )

gap> DirectedEdges( gamma );
[ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 2, 1 ], [ 2, 3 ], [ 2, 4 ], [ 3, 1 ],
[ 3, 2 ], [ 3, 4 ], [ 4, 1 ], [ 4, 2 ], [ 4, 3 ] ]

gap> UndirectedEdges( gamma );
[ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 2, 3 ], [ 2, 4 ], [ 3, 4 ] ]

3.14 UndirectedEdges

1I UndirectedEdges( gamma )

This function returns the set of undirected (unordered) edges of gamma, which must be a simple graph.

See also 3.13.1.

gap> gamma := JohnsonGraph( 4, 3 );
rec( isGraph := true, order := 4, group := Group([ (1,4,3,2), (3,4) ]),
schreierVector := [ -1, 1, 1, 1 ], adjacencies := [ [ 2, 3, 4 ] ],
representatives := [ 1 ],
names := [ [ 1, 2, 3 ], [ 1, 2, 4 ], [ 1, 3, 4 ], [ 2, 3, 4 ] ],
isSimple := true )

gap> DirectedEdges( gamma );
[ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 2, 1 ], [ 2, 3 ], [ 2, 4 ], [ 3, 1 ],
[ 3, 2 ], [ 3, 4 ], [ 4, 1 ], [ 4, 2 ], [ 4, 3 ] ]

gap> UndirectedEdges( gamma );
[ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 2, 3 ], [ 2, 4 ], [ 3, 4 ] ]

3.15 Distance

1I Distance( gamma, X , Y )
I Distance( gamma, X , Y , G )

This function returns the distance from X to Y in gamma. The parameters X and Y may be vertices or
nonempty lists of vertices. We define the distance d(X ,Y ) from X to Y to be the minimum length of a
(directed) path joining a vertex of X to a vertex of Y if such a path exists, and −1 otherwise.

The optional parameter G , if present, is assumed to be a subgroup of Aut (gamma) fixing X setwise.
Including such a G can speed up the function.

See also 3.16.1.

gap> Distance( JohnsonGraph(4,2), 1, 6 );
2
gap> Distance( JohnsonGraph(4,2), 1, 5 );
1
gap> Distance( JohnsonGraph(4,2), [1], [5,6] );
1

3.16 Diameter

1I Diameter( gamma )

This function returns the diameter of gamma. A diameter of −1 is returned if gamma is not (strongly)
connected. Otherwise, the diameter of gamma is equal to the maximum (directed) distance d(x , y) in
gamma (as x and y range over all the vertices of gamma).

See also 3.15.1.

gap> Diameter( JohnsonGraph( 5, 3 ) );
2
gap> Diameter( JohnsonGraph( 5, 4 ) );
1

3.17 Girth

1I Girth( gamma )

This function returns the girth of gamma, which must be a simple graph. A girth of −1 is returned if gamma
is a forest. Otherwise the girth is the length of a shortest cycle in gamma.

gap> Girth( JohnsonGraph( 4, 2 ) );
3

3.18 IsConnectedGraph

1I IsConnectedGraph( gamma )

This boolean function returns true if and only if the graph gamma is (strongly) connected, i.e. there is a
(directed) path from x to y for every pair of vertices x , y of gamma.

gap> IsConnectedGraph( JohnsonGraph(4,2) );
true
gap> IsConnectedGraph( NullGraph(SymmetricGroup(4)) );
false

3.19 IsBipartite

1I IsBipartite( gamma )

This boolean function returns true if and only if the graph gamma, which must be simple, is bipartite,
i.e. if the vertex-set can be expressed as the disjoint union of two sets, on each of which gamma induces a
null graph (these two sets are called the bicomponents or parts of gamma).

See also 5.3.1 and 6.10.1.

gap> gamma := JohnsonGraph(4,2);
rec(
isGraph := true,
order := 6,
group := Group( [ (1,4,6,3)(2,5), (2,4)(3,5) ] ),
schreierVector := [ -1, 2, 1, 1, 1, 1 ],
adjacencies := [ [ 2, 3, 4, 5 ] ],
representatives := [ 1 ],
names := [ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 2, 3 ], [ 2, 4 ], [ 3, 4 ] ],
isSimple := true )

gap> IsBipartite(gamma);
false
gap> delta := BipartiteDouble(gamma);
rec(
isGraph := true,
order := 12,
group := Group( [ ( 1, 4, 6, 3)( 2, 5)( 7,10,12, 9)( 8,11),

( 2, 4)( 3, 5)( 8,10)( 9,11), ( 1, 7)( 2, 8)( 3, 9)( 4,10)( 5,11)
( 6,12) ] ),

schreierVector := [ -1, 2, 1, 1, 1, 1, 3, 3, 3, 3, 3, 3 ],
adjacencies := [ [ 8, 9, 10, 11 ] ],
representatives := [ 1 ],
isSimple := true,
names := [ [ [ 1, 2 ], "+" ], [ [ 1, 3 ], "+" ], [ [ 1, 4 ], "+" ],

[ [ 2, 3 ], "+" ], [ [ 2, 4 ], "+" ], [ [ 3, 4 ], "+" ],
[ [ 1, 2 ], "-" ], [ [ 1, 3 ], "-" ], [ [ 1, 4 ], "-" ],
[ [ 2, 3 ], "-" ], [ [ 2, 4 ], "-" ], [ [ 3, 4 ], "-" ] ] )

gap> IsBipartite(delta);
true

3.20 IsNullGraph

1I IsNullGraph( gamma )

This boolean function returns true if and only if the graph gamma has no edges.

See also 2.3.1.

gap> IsNullGraph( CompleteGraph( Group(()), 3 ) );
false
gap> IsNullGraph( CompleteGraph( Group(()), 1 ) );
true

3.21 IsCompleteGraph

1I IsCompleteGraph( gamma )
I IsCompleteGraph( gamma, mustloops )

This boolean function returns true if and only if the graph gamma is a complete graph. The optional
boolean parameter mustloops determines whether all loops must be present for gamma to be considered a
complete graph (default: false (loops are ignored)).

See also 2.4.1.

gap> IsCompleteGraph( NullGraph( Group(()), 3 ) );
false
gap> IsCompleteGraph( NullGraph( Group(()), 1 ) );
true
gap> IsCompleteGraph( CompleteGraph(SymmetricGroup(3)), true );
false

Functions to
determine regularity
properties of graphs

This chapter describes functions to determine regularity properties of graphs, and a function VertexTran-
sitiveDRGs which determines the distance-regular graphs on which a given transitive permutation group
acts as a vertex-transitive group of automorphisms.

4.1 IsRegularGraph

1I IsRegularGraph( gamma )

This boolean function returns true if and only if the graph gamma is (out)regular.

gap> IsRegularGraph( JohnsonGraph(4,2) );
true
gap> IsRegularGraph( EdgeOrbitsGraph(Group(()),[[1,2]],2) );
false

4.2 LocalParameters

1I LocalParameters( gamma, V )
I LocalParameters( gamma, V , G )

Let gamma be a simple connected graph. Then this function determines all local parameters ci (V ), ai (V ),
and bi (V ) that gamma may have, with respect to the singleton vertex or nonempty list of vertices V . We say
that gamma has the local parameter ci (V ) (respectively ai (V ), bi (V )), with respect to V , if the number
of vertices at distance i − 1 (respectively i , i + 1) from V that are adjacent to a vertex w at distance i from
V (see 3.15.1) is the constant ci (V ) (respectively ai (V ), bi (V )) depending only on i and V (and not w).

The function LocalParameters returns a list whose i -th element is the list [ci−1(V ), ai−1(V ), bi−1(V )],
except that if some local parameter does not exist then −1 is put in its place.

This function can be used to determine whether a given subset of the vertices of a graph is a distance-regular
code in that graph.

The optional parameter G , if present, is assumed to be a subgroup of Aut (gamma) fixing V setwise.
Including such a G can speed up the function.

gap> gamma := JohnsonGraph(4,2);;
gap> LocalParameters( gamma, 1 );
[ [ 0, 0, 4 ], [ 1, 2, 1 ], [ 4, 0, 0 ] ]
gap> LocalParameters( gamma, [1,6] );
[ [ 0, 0, 4 ], [ 2, 2, 0 ] ]
gap> LocalParameters( gamma, [1,2] );
[ [ 0, 1, 3 ], [ -1, -1, 0 ] ]

4.3 GlobalParameters

1I GlobalParameters( gamma )

Let gamma be a simple connected graph, and 0 ≤ i ≤ Diameter(gamma). This function determines all
global parameters ci , ai , and bi that gamma may have. We say that gamma has the global parameter ci
(respectively ai , bi ) if the number of vertices at distance i − 1 (respectively i , i + 1) from a vertex v that
are adjacent to a vertex w at distance i from v is the constant ci (respectively ai , bi ) depending only on i
(and not v and w).

The function GlobalParameters returns a list of length Diameter(gamma)+1, whose i -th element is the
list [ci−1, ai−1, bi−1], except that if some global parameter does not exist then −1 is put in its place.

Note that gamma is distance-regular if and only if this function returns no −1 in place of a global
parameter (see [BCN89]).

See also 4.2.1 and 4.4.1.

gap> gamma := JohnsonGraph(4,2);;
gap> GlobalParameters( gamma );
[ [ 0, 0, 4 ], [ 1, 2, 1 ], [ 4, 0, 0 ] ]
gap> GlobalParameters( BipartiteDouble(gamma) );
[ [ 0, 0, 4 ], [ 1, 0, 3 ], [ -1, 0, -1 ], [ 4, 0, 0 ] ]

4.4 IsDistanceRegular

1I IsDistanceRegular( gamma )

This boolean function returns true if and only if gamma is distance-regular, i.e. gamma is simple, con-
nected, and all global parameters ci , ai , bi exist for 0 ≤ i ≤ Diameter(gamma) (see [BCN89]).

See also 4.3.1.

gap> gamma := JohnsonGraph(4,2);;
gap> IsDistanceRegular( gamma );
true
gap> IsDistanceRegular( BipartiteDouble(gamma) );
false

4.5 CollapsedAdjacencyMat

1I CollapsedAdjacencyMat( gamma )
I CollapsedAdjacencyMat( G, gamma )

The second form of this function returns the collapsed adjacency matrix for gamma, where the collapsing
group is G . It is assumed that G is a subgroup of Aut (gamma).

The (i , j )-entry of the collapsed adjacency matrix equals the number of edges in {[x , y ] | y ∈ j -th G-orbit},
where x is a fixed vertex in the i -th G-orbit.

In the case where this function is given just one argument, then it must be a graph gamma with the property
that gamma.group is transitive on the vertex-set of gamma. In this case, the returned collapsed adjacency
matrix for gamma is with respect to the stabilizer in gamma.group of 1.

See also 4.6.1.

gap> gamma := JohnsonGraph(4,2);
rec( isGraph := true, order := 6,
group := Group([ (1,4,6,3)(2,5), (2,4)(3,5) ]),
schreierVector := [ -1, 2, 1, 1, 1, 1 ], adjacencies := [ [ 2, 3, 4, 5 ] ],
representatives := [ 1 ],
names := [ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 2, 3 ], [ 2, 4 ], [ 3, 4 ] ],
isSimple := true )

gap> G := Stabilizer( gamma.group, [1,6], OnSets );;
gap> CollapsedAdjacencyMat( G, gamma );
[ [ 0, 4 ], [ 2, 2 ] ]
gap> CollapsedAdjacencyMat( gamma );
[ [ 0, 4, 0 ], [ 1, 2, 1 ], [ 0, 4, 0 ] ]

4.6 OrbitalGraphColadjMats

1I OrbitalGraphColadjMats( G )
I OrbitalGraphColadjMats( G, H )

This function returns a list of collapsed adjacency matrices for the orbital digraphs of the transitive permu-
tation group G , collapsed with respect to Stabilizer(G,1) (creating collapsed adjacency matrices for the
orbital digraphs in the sense of [PS97]). Also, the matrices are collapsed with respect to a fixed ordering of
the orbits of Stabilizer(G,1), with the trivial orbit [1] coming first.

The optional parameter H , if included, should be equal to Stabilizer(G,1). The knowledge of this stabi-
lizer can speed up the function.

For backward compatibility, an alternative name for this function is OrbitalGraphIntersectionMatri-
ces, but the reader is warned that intersection matrices (and even collapsed adjacency matrices) can have
different, but closely related meanings depending on the setting and the author. See, for example, [Cam99].

See also 4.5.1.

gap> OrbitalGraphColadjMats( SymmetricGroup(7) );
[ [ [ 1, 0 ], [ 0, 1 ] ], [ [ 0, 6 ], [ 1, 5 ] ] ]

4.7 VertexTransitiveDRGs

1I VertexTransitiveDRGs( coladjmats )
I VertexTransitiveDRGs( G )

This function can determine (among other things) all the distance-regular graphs on which a given transitive
permutation group G acts as a vertex-transitive group of automorphisms (as long as the permutation rank
of G is not too large).

In the first form of this function, the input parameter coladjmats must be a list of collapsed adjacency
matrices for the orbital digraphs of some transitive permutation group G , collapsed with respect to a point
stabilizer (such as the list of matrices produced by the function OrbitalGraphColadjMats). It is assumed
that the orbital/suborbit indexing used is the same as that for the rows (and columns) of each of the
matrices, as well as for the indexing of the matrices themselves, with the trivial orbital first, so that, in
particular, coladjmats[1] must be an identity matrix.

In the second form of this function, the input parameter G must be a transitive permutation group, and
then the result returned will be the same as VertexTransitiveDRGs( OrbitalGraphColadjMats( G ) ).

In either case, this function returns a record result , which gives information on the transitive group G acting
on its natural set V . The most important component of this record is the list orbitalCombinations, whose
elements give the sets of (the indices of) the G-orbitals whose union gives the edge-set of a distance-regular

graph with vertex-set V . The component intersectionArrays gives the corresponding intersection arrays.
The component degree is the degree of the permutation group G , rank is its (permutation) rank, and
isPrimitive is true if G is primitive, and false otherwise.

The techniques used in this function and definitions of the terms used above can be found in [PS97].

Warning This function checks all subsets of [2..result.rank], so the permutation rank of G must not be
large!

gap> m22:=PrimitiveGroup(22,1);;
gap> syl:=SylowSubgroup(m22,11);;
gap> part:=Set(Orbit(syl,1));;
gap> l211:=Stabilizer(m22,part,OnSets);;
gap> rt:=RightTransversal(m22,l211);;
gap> m22big:=Action(m22,rt,OnRight);;
gap> v:=VertexTransitiveDRGs(m22big);
rec( degree := 672, rank := 6, isPrimitive := true,
orbitalCombinations := [ [ 2, 3, 4, 5, 6 ], [ 2, 4 ], [ 3, 5, 6 ], [ 3, 6 ]

],
intersectionArrays := [ [ [ 0, 0, 671 ], [ 1, 670, 0 ] ], [ [ 0, 0, 495 ],

[ 1, 366, 128 ], [ 360, 135, 0 ] ],
[ [ 0, 0, 176 ], [ 1, 40, 135 ], [ 48, 128, 0 ] ],
[ [ 0, 0, 110 ], [ 1, 28, 81 ], [ 18, 80, 12 ], [ 90, 20, 0 ] ] ] )

Some special vertex
subsets of a graph

This chapter describes functions to determine certain special vertex subsets of a graph.

5.1 ConnectedComponent

1I ConnectedComponent( gamma, v )

This function returns the set of all vertices in gamma which can be reached by a path starting at the vertex
v . The graph gamma must be simple.

See also 5.2.1.

gap> ConnectedComponent( NullGraph( Group((1,2)) ), 2 );
[ 2 ]
gap> ConnectedComponent( JohnsonGraph(4,2), 2 );
[ 1, 2, 3, 4, 5, 6 ]

5.2 ConnectedComponents

1I ConnectedComponents( gamma )

This function returns a list of the vertex sets of the connected components of gamma, which must be a
simple graph.

See also 5.1.1.

gap> ConnectedComponents( NullGraph( Group((1,2,3,4)) ) );
[ [ 1 ], [ 2 ], [ 3 ], [ 4 ] ]
gap> ConnectedComponents( JohnsonGraph(4,2) );
[ [ 1, 2, 3, 4, 5, 6 ] ]

5.3 Bicomponents

1I Bicomponents( gamma )

If the graph gamma, which must be simple, is bipartite, this function returns a length 2 list of bicomponents
or parts of gamma, otherwise the empty list is returned.

Note If gamma is bipartite but not connected, then its set of bicomponents is not uniquely determined.

See also 3.19.1.

gap> Bicomponents( NullGraph(SymmetricGroup(4)) );
[ [ 1 .. 3 ], [ 4 ] ]
gap> Bicomponents( JohnsonGraph(4,2) );
[ ]
gap> Bicomponents( BipartiteDouble( JohnsonGraph(4,2) ) );
[ [ 1, 2, 3, 4, 5, 6 ], [ 7, 8, 9, 10, 11, 12 ] ]

5.4 DistanceSet

1I DistanceSet( gamma, distances, V )
I DistanceSet( gamma, distances, V , G )

Let V be a vertex or a nonempty list of vertices of gamma. This function returns the set of vertices w of
gamma, such that d(V ,w) is in distances (a list or singleton distance).

The optional parameter G , if present, is assumed to be a subgroup of Aut (gamma) fixing V setwise.
Including such a G can speed up the function.

See also 3.15.1 and 6.2.1.

gap> DistanceSet( JohnsonGraph(4,2), 1, [1,6] );
[ 2, 3, 4, 5 ]

5.5 Layers

1I Layers( gamma, V )
I Layers( gamma, V , G )

Let V be a vertex or a nonempty list of vertices of gamma. This function returns a list whose i -th element
is the set of vertices of gamma at distance i − 1 from V .

The optional parameter G , if present, is assumed to be a subgroup of Aut (gamma) which fixes V setwise.
Including such a G can speed up the function.

See also 3.15.1.

gap> Layers( JohnsonGraph(4,2), 6 );
[ [ 6 ], [ 2, 3, 4, 5 ], [ 1 ] ]

5.6 IndependentSet

1I IndependentSet( gamma )
I IndependentSet( gamma, indset )
I IndependentSet( gamma, indset, forbidden )

Returns a (hopefully large) independent set of the graph gamma, which must be simple. An independent
set of gamma is a set of vertices of gamma, no two of which are joined by an edge. At present, a greedy
algorithm is used. The returned independent set will contain the (assumed) independent set indset (default:
[]), and not contain any element of forbidden (default: [], in which case the returned independent set is
maximal).

An error is signalled if indset and forbidden have non-trivial intersection.

See also 7.2.1 and 7.3.1, which can be used on the complement graph of gamma to look seriously for
independent sets.

gap> IndependentSet( JohnsonGraph(4,2), [3] );
[ 3, 4 ]

Functions to construct
new graphs from old

This chapter describes functions to construct new graphs from old ones.

6.1 InducedSubgraph

1I InducedSubgraph( gamma, V )
I InducedSubgraph( gamma, V , G )

This function returns the subgraph of gamma induced on the vertex list V (which must not contain repeated
elements). If the optional third parameter G is given, then it is assumed that G fixes V setwise, and is a
group of automorphisms of the induced subgraph when restricted to V . In that case, the image of G acting
on V is the group associated with the induced subgraph. If no such G is given then the associated group is
trivial. The name of vertex i in the induced subgraph is equal to the name of vertex V [i] in gamma.

gap> gamma := JohnsonGraph(4,2);;
gap> S := [2,3,4,5];;
gap> square := InducedSubgraph( gamma, S, Stabilizer(gamma.group,S,OnSets) );
rec(
isGraph := true,
order := 4,
group := Group( [ (1,4), (1,3)(2,4), (1,2)(3,4) ] ),
schreierVector := [ -1, 3, 2, 1 ],
adjacencies := [ [ 2, 3 ] ],
representatives := [ 1 ],
isSimple := true,
names := [ [ 1, 3 ], [ 1, 4 ], [ 2, 3 ], [ 2, 4 ] ] )

gap> GlobalParameters(square);
[ [ 0, 0, 2 ], [ 1, 0, 1 ], [ 2, 0, 0 ] ]

6.2 DistanceSetInduced

1I DistanceSetInduced( gamma, distances, V )
I DistanceSetInduced( gamma, distances, V , G )

Let V be a vertex or a nonempty list of vertices of gamma. This function returns the subgraph of gamma
induced on the set of vertices w of gamma such that d(V ,w) is in distances (a list or singleton distance).

The optional parameter G , if present, is assumed to be a subgroup of Aut (gamma) fixing V setwise.
Including such a G can speed up the function.

See also 3.15.1 and 5.4.1.

gap> DistanceSetInduced( JohnsonGraph(4,2), [0,1], [1] );
rec(
isGraph := true,
order := 5,
group := Group( [ (2,3)(4,5), (2,5)(3,4) ] ),
schreierVector := [ -1, -2, 1, 2, 2 ],
adjacencies := [ [ 2, 3, 4, 5 ], [ 1, 3, 4 ] ],
representatives := [ 1, 2 ],
isSimple := true,
names := [ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 2, 3 ], [ 2, 4 ] ] )

6.3 DistanceGraph

1I DistanceGraph( gamma, distances )

This function returns the graph delta, with the same vertex-set (and vertex-names) as gamma, such that
[x , y ] is an edge of delta if and only if d(x , y) (in gamma) is in distances (a list or singleton distance).

gap> DistanceGraph( JohnsonGraph(4,2), [2] );
rec(
isGraph := true,
order := 6,
group := Group( [ (1,4,6,3)(2,5), (2,4)(3,5) ] ),
schreierVector := [ -1, 2, 1, 1, 1, 1 ],
adjacencies := [ [ 6 ] ],
representatives := [ 1 ],
names := [ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 2, 3 ], [ 2, 4 ], [ 3, 4 ] ],
isSimple := true )

gap> ConnectedComponents(last);
[ [ 1, 6 ], [ 2, 5 ], [ 3, 4 ] ]

6.4 ComplementGraph

1I ComplementGraph( gamma )
I ComplementGraph( gamma, comploops )

This function returns the complement of the graph gamma. The optional boolean parameter comploops
determines whether or not loops/nonloops are complemented (default: false (loops/nonloops are not com-
plemented)). The returned graph will have the same vertex-names as gamma.

gap> ComplementGraph( NullGraph(SymmetricGroup(3)) );
rec(
isGraph := true,
order := 3,
group := SymmetricGroup( [ 1 .. 3 ] ),
schreierVector := [ -1, 1, 1 ],
adjacencies := [ [ 2, 3 ] ],
representatives := [ 1 ],
isSimple := true )

gap> IsLoopy(last);
false
gap> IsLoopy(ComplementGraph(NullGraph(SymmetricGroup(3)),true));
true

6.5 PointGraph

1I PointGraph( gamma )
I PointGraph( gamma, v )

Assuming that gamma is simple, connected, and bipartite, this function returns the induced subgraph on
the connected component of DistanceGraph(gamma,2) containing the vertex v (default: v = 1).

Thus, if gamma is the incidence graph of a connected geometry of rank 2, and v represents a point, then
the point graph of the geometry is returned.

gap> BipartiteDouble( CompleteGraph(SymmetricGroup(4)) );;
gap> PointGraph(last);
rec(
isGraph := true,
order := 4,
group := Group( [ (1,2), (1,2,3,4) ] ),
schreierVector := [ -1, 1, 2, 2 ],
adjacencies := [ [ 2, 3, 4 ] ],
representatives := [ 1 ],
isSimple := true,
names := [ [ 1, "+" ], [ 2, "+" ], [ 3, "+" ], [ 4, "+" ] ] )

gap> IsCompleteGraph(last);
true

6.6 EdgeGraph

1I EdgeGraph( gamma )

This function return a graph isomorphic to the the edge graph (also called the line graph) of the simple
graph gamma.

This edge graph delta has the unordered edges of gamma as vertices, and e is joined to f in delta precisely
when |e ∩ f | = 1. The name of the vertex of the returned graph corresponding to the unordered edge [v ,w ]
of gamma (with v < w) is [VertexName(gamma,v),VertexName(gamma,w)].

gap> EdgeGraph( CompleteGraph(SymmetricGroup(5)) );
rec(
isGraph := true,
order := 10,
group := Group( [ ( 1, 5, 8,10, 4)( 2, 6, 9, 3, 7), ( 2, 5)( 3, 6)( 4, 7)

] ),
schreierVector := [ -1, 2, 2, 1, 1, 1, 2, 1, 1, 1 ],
adjacencies := [ [ 2, 3, 4, 5, 6, 7 ] ],
representatives := [ 1 ],
isSimple := true,
names := [ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 1, 5 ], [ 2, 3 ], [ 2, 4 ],

[ 2, 5 ], [ 3, 4 ], [ 3, 5 ], [ 4, 5 ] ] )
gap> GlobalParameters(last);
[ [ 0, 0, 6 ], [ 1, 3, 2 ], [ 4, 2, 0 ] ]

6.7 SwitchedGraph

1I SwitchedGraph( gamma, V )
I SwitchedGraph( gamma, V , H )

This function returns the switched graph delta of the graph gamma, switched with respect to the vertex list
(or singleton vertex) V .

The returned graph delta has vertex-set (and vertex-names) the same as gamma. If vertices x , y of delta are
both in V or both not in V , then [x , y ] is an edge of delta if and only if [x , y ] is an edge of gamma; otherwise
[x , y ] is an edge of delta if and only if [x , y ] is not an edge of gamma. If the optional third argument H is
given, then it is assumed to be a subgroup of Aut(gamma) stabilizing V setwise.

gap> J:=JohnsonGraph(4,2);
rec(
isGraph := true,
order := 6,
group := Group( [ (1,4,6,3)(2,5), (2,4)(3,5) ] ),
schreierVector := [ -1, 2, 1, 1, 1, 1 ],
adjacencies := [ [ 2, 3, 4, 5 ] ],
representatives := [ 1 ],
names := [ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 2, 3 ], [ 2, 4 ], [ 3, 4 ] ],
isSimple := true )

gap> S:=SwitchedGraph(J,[1,6]);
rec(
isGraph := true,
order := 6,
group := Group( () ),
schreierVector := [ -1, -2, -3, -4, -5, -6 ],
adjacencies := [ [ ], [ 3, 4 ], [ 2, 5 ], [ 2, 5 ], [ 3, 4 ], [ ] ],
representatives := [ 1, 2, 3, 4, 5, 6 ],
isSimple := true,
names := [ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 2, 3 ], [ 2, 4 ], [ 3, 4 ] ] )

gap> ConnectedComponents(S);
[ [ 1 ], [ 2, 3, 4, 5 ], [ 6 ] ]

6.8 UnderlyingGraph

1I UnderlyingGraph( gamma )

This function returns the underlying graph delta of gamma. The graph delta has the same vertex-set (and
vertex-names) as gamma, and has an edge [x , y ] precisely when gamma has an edge [x , y ] or an edge [y , x ].
This function also sets the isSimple components of gamma and delta.

gap> gamma := EdgeOrbitsGraph( Group((1,2,3,4)), [1,2] );
rec(
isGraph := true,
order := 4,
group := Group( [ (1,2,3,4) ] ),
schreierVector := [ -1, 1, 1, 1 ],
adjacencies := [ [ 2 ] ],
representatives := [ 1 ],
isSimple := false )

gap> UnderlyingGraph(gamma);
rec(



Section 10. BipartiteDouble 31

isGraph := true,
order := 4,
group := Group( [ (1,2,3,4) ] ),
schreierVector := [ -1, 1, 1, 1 ],
adjacencies := [ [ 2, 4 ] ],
representatives := [ 1 ],
isSimple := true )

6.9 QuotientGraph

1I QuotientGraph( gamma, R )

Let S be the smallest gamma.group-invariant equivalence relation on the vertices of gamma, such that S
contains the relation R (which should be a list of ordered pairs (length 2 lists) of vertices of gamma). Then
this function returns a graph isomorphic to the quotient delta of the graph gamma, defined as follows. The
vertices of delta are the equivalence classes of S , and [X ,Y ] is an edge of delta if and only if [x , y ] is an edge
of gamma for some x ∈ X , y ∈ Y . The name of a vertex v in the returned graph is a list (not necessarily
ordered) of the vertex-names of gamma for the vertices in the equivalence class corresponding to v .

gap> gamma := JohnsonGraph(4,2);;
gap> QuotientGraph( gamma, [[1,6]] );
rec(
isGraph := true,
order := 3,
group := Group( [ (1,3), (2,3) ] ),
schreierVector := [ -1, 2, 1 ],
adjacencies := [ [ 2, 3 ] ],
representatives := [ 1 ],
isSimple := true,
names := [ [ [ 1, 2 ], [ 3, 4 ] ], [ [ 1, 3 ], [ 2, 4 ] ],

[ [ 1, 4 ], [ 2, 3 ] ] ] )
gap> IsCompleteGraph(last);
true

6.10 BipartiteDouble

1I BipartiteDouble( gamma )

This function returns the bipartite double of the graph gamma, as defined in [BCN89].

gap> gamma := JohnsonGraph(4,2);;
gap> IsBipartite(gamma);
false
gap> delta := BipartiteDouble(gamma);
rec(
isGraph := true,
order := 12,
group := Group( [ ( 1, 4, 6, 3)( 2, 5)( 7,10,12, 9)( 8,11),

( 2, 4)( 3, 5)( 8,10)( 9,11), ( 1, 7)( 2, 8)( 3, 9)( 4,10)( 5,11)
( 6,12) ] ),

schreierVector := [ -1, 2, 1, 1, 1, 1, 3, 3, 3, 3, 3, 3 ],
adjacencies := [ [ 8, 9, 10, 11 ] ],
representatives := [ 1 ],
isSimple := true,

names := [ [ [ 1, 2 ], "+" ], [ [ 1, 3 ], "+" ], [ [ 1, 4 ], "+" ],
[ [ 2, 3 ], "+" ], [ [ 2, 4 ], "+" ], [ [ 3, 4 ], "+" ],
[ [ 1, 2 ], "-" ], [ [ 1, 3 ], "-" ], [ [ 1, 4 ], "-" ],
[ [ 2, 3 ], "-" ], [ [ 2, 4 ], "-" ], [ [ 3, 4 ], "-" ] ] )

gap> IsBipartite(delta);
true

6.11 GeodesicsGraph

1I GeodesicsGraph( gamma, x, y )

This function returns the the graph induced on the set of geodesics in gamma between the vertices x and
y , but including neither x nor y . This function is only for a simple graph gamma.

gap> GeodesicsGraph( JohnsonGraph(4,2), 1, 6 );
rec(
isGraph := true,
order := 4,
group := Group( [ (1,3)(2,4), (1,4)(2,3), (2,3) ] ),
schreierVector := [ -1, 2, 1, 2 ],
adjacencies := [ [ 2, 3 ] ],
representatives := [ 1 ],
isSimple := true,
names := [ [ 1, 3 ], [ 1, 4 ], [ 2, 3 ], [ 2, 4 ] ] )

gap> GlobalParameters(last);
[ [ 0, 0, 2 ], [ 1, 0, 1 ], [ 2, 0, 0 ] ]

6.12 CollapsedIndependentOrbitsGraph

1I CollapsedIndependentOrbitsGraph( G, gamma )
I CollapsedIndependentOrbitsGraph( G, gamma, N )

Given a subgroup G of the automorphism group of the simple graph gamma, this function returns a graph
isomorphic to delta, defined as follows. The vertices of delta are those G-orbits of the vertices of gamma that
are independent sets in gamma, and x is joined to y in delta if and only if x ∪y is not an independent set in
gamma. The name of a vertex v in the returned graph is a list (not necessarily ordered) of the vertex-names
of gamma for the vertices in the G-orbit corresponding to v .

If the optional parameter N is given, then it is assumed to be a subgroup of Aut (gamma) preserving the set
of G-orbits of the vertices of gamma (for example, the normalizer in gamma.group of G). This information
can make the function more efficient.

gap> G := Group( (1,2) );;
gap> gamma := NullGraph( SymmetricGroup(3) );;
gap> CollapsedIndependentOrbitsGraph( G, gamma );
rec(
isGraph := true,
order := 2,
group := Group( [ () ] ),
schreierVector := [ -1, -2 ],
adjacencies := [ [ ], [ ] ],
representatives := [ 1, 2 ],
isSimple := true,
names := [ [ 1, 2 ], [ 3 ] ] )

gap> gamma := CompleteGraph( SymmetricGroup(3) );;
gap> CollapsedIndependentOrbitsGraph( G, gamma );
rec(
isGraph := true,
order := 1,
group := Group( [ () ] ),
schreierVector := [ -1 ],
adjacencies := [ [ ] ],
representatives := [ 1 ],
isSimple := true,
names := [ [ 3 ] ] )

6.13 CollapsedCompleteOrbitsGraph

1I CollapsedCompleteOrbitsGraph( G, gamma )
I CollapsedCompleteOrbitsGraph( G, gamma, N )

Given a subgroup G of the automorphism group of the simple graph gamma, this function returns a graph
isomorphic to delta, defined as follows. The vertices of delta are those G-orbits of the vertices of gamma on
which complete subgraphs are induced in gamma, and x is joined to y in delta if and only if x 6= y and the
subgraph of gamma induced on x ∪ y is a complete graph. The name of a vertex v in the returned graph is
a list (not necessarily ordered) of the vertex-names of gamma for the vertices in the G-orbit corresponding
to v .

If the optional parameter N is given, then it is assumed to be a subgroup of Aut (gamma) preserving the set
of G-orbits of the vertices of gamma (for example, the normalizer in gamma.group of G). This information
can make the function more efficient.

gap> G := Group( (1,2) );;
gap> gamma := NullGraph( SymmetricGroup(3) );;
gap> CollapsedCompleteOrbitsGraph( G, gamma );
rec(
isGraph := true,
order := 1,
group := Group( [ () ] ),
schreierVector := [ -1 ],
adjacencies := [ [ ] ],
representatives := [ 1 ],
names := [ [ 3 ] ],
isSimple := true )

gap> gamma := CompleteGraph( SymmetricGroup(3) );;
gap> CollapsedCompleteOrbitsGraph( G, gamma );
rec(
isGraph := true,
order := 2,
group := Group( [ () ] ),
schreierVector := [ -1, -2 ],
adjacencies := [ [ 2 ], [ 1 ] ],
representatives := [ 1, 2 ],
names := [ [ 1, 2 ], [ 3 ] ],
isSimple := true )

6.14 NewGroupGraph

1I NewGroupGraph( G, gamma )

This function returns a copy delta of gamma, except that the group associated with delta is G , which is
assumed to be a subgroup of Aut (delta).

Note that the results of some functions of a graph depend on the group associated with that graph (which
must always be a subgroup of the automorphism group of the graph).

gap> gamma := JohnsonGraph(4,2);;
gap> aut := AutGroupGraph(gamma);
Group([ (3,4), (2,3)(4,5), (1,2)(5,6) ])
gap> Size(gamma.group);
24
gap> Size(aut);
48
gap> delta := NewGroupGraph( aut, gamma );;
gap> Size(delta.group);
48
gap> IsIsomorphicGraph( gamma, delta );
true

Vertex-Colouring and
Complete Subgraphs

The following sections describe functions for (proper) vertex-colouring or determining complete subgraphs
of given graphs. The function CompleteSubgraphsOfGivenSize can also be used to determine the complete
subgraphs with given vertex-weight sum in a vertex-weighted graph.

7.1 VertexColouring

1I VertexColouring( gamma )

This function returns a proper vertex-colouring C for the graph gamma, which must be simple.

This proper vertex-colouring C is a list of positive integers, indexed by the vertices of gamma, with the
property that C [i ] 6= C [j ] whenever [i , j ] is an edge of gamma. At present a greedy algorithm is used.

gap> VertexColouring( JohnsonGraph(4,2) );
[ 1, 3, 2, 2, 3, 1 ]

7.2 CompleteSubgraphs

1I CompleteSubgraphs( gamma )
I CompleteSubgraphs( gamma, k )
I CompleteSubgraphs( gamma, k, alls )

Let gamma be a simple graph and k an integer. This function returns a set K of complete subgraphs of
gamma, where a complete subgraph is represented by its vertex-set. If k is non-negative then the elements
of K each have size k , otherwise the elements of K represent maximal complete subgraphs of gamma. The
default for k is −1, i.e. maximal complete subgraphs. See also CompleteSubgraphsOfGivenSize, which can
be used to compute the maximal complete subgraphs of given size, and can also be used to determine the
(maximal or otherwise) complete subgraphs with given vertex-weight sum in a vertex-weighted graph.

The optional parameter alls controls how many complete subgraphs are returned. The valid values for alls
are 0, 1 (the default), and 2. The value 2 provides a new feature in GRAPE from version 4.1, and specifies
that this function should compute a set of gamma.group-orbit representatives of the required complete
subgraphs.

If alls=0 (or false for backward compatibility) then K will contain at most one element. In this case, if k
is negative then K will contain just one maximal complete subgraph, and if k is non-negative then K will
contain a complete subgraph of size k if and only if such a subgraph is contained in gamma.

If alls=1 (or true for backward compatibility) then K will contain (perhaps properly) a set of gamma.group
orbit-representatives of the maximal (if k is negative) or size k (if k is non-negative) complete subgraphs of
gamma.

If alls=2 then K will be a set of gamma.group orbit-representatives of the maximal (if k is negative) or size
k (if k is non-negative) complete subgraphs of gamma. This option can be more costly than when alls=1.

Before applying CompleteSubgraphs, one may want to associate the full automorphism group of gamma
with gamma, via gamma := NewGroupGraph( AutGroupGraph(gamma), gamma );.

An alternative name for this function is Cliques.

See also 7.3.1.

gap> gamma := JohnsonGraph(5,2);
rec( isGraph := true, order := 10,
group := Group([ ( 1, 5, 8,10, 4)( 2, 6, 9, 3, 7), ( 2, 5)( 3, 6)( 4, 7) ]),
schreierVector := [ -1, 2, 2, 1, 1, 1, 2, 1, 1, 1 ],
adjacencies := [ [ 2, 3, 4, 5, 6, 7 ] ], representatives := [ 1 ],
names := [ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 1, 5 ], [ 2, 3 ], [ 2, 4 ],

[ 2, 5 ], [ 3, 4 ], [ 3, 5 ], [ 4, 5 ] ], isSimple := true )
gap> CompleteSubgraphs(gamma);
[ [ 1, 2, 3, 4 ], [ 1, 2, 5 ] ]
gap> CompleteSubgraphs(gamma,3,2);
[ [ 1, 2, 3 ], [ 1, 2, 5 ] ]
gap> CompleteSubgraphs(gamma,-1,0);
[ [ 1, 2, 5 ] ]

7.3 CompleteSubgraphsOfGivenSize

1I CompleteSubgraphsOfGivenSize( gamma, k )
I CompleteSubgraphsOfGivenSize( gamma, k, alls )
I CompleteSubgraphsOfGivenSize( gamma, k, alls, maxi )
I CompleteSubgraphsOfGivenSize( gamma, k, alls, maxi, col )
I CompleteSubgraphsOfGivenSize( gamma, k, alls, maxi, col, wts )

Let gamma be a simple graph and k a non-negative integer. This function returns a set K (possibly empty)
of complete subgraphs of size k of gamma. The exact nature of the set K depends on the values of the
parameters supplied to this function. A complete subgraph is represented by its vertex-set. CompleteSub-
graphsOfGivenSize can handle the case where the vertices are assigned positive integer weights, in which
case, a complete subgraph of size k means a complete subgraph whose vertex-weights sum to k .

We remark that the parameter maxi provides a backward compatible extension to its use in versions of
GRAPE previous to 4.1, and now when maxi=true this function returns only maximal complete subgraphs
of size k . The parameter alls also provides a new feature starting with version 4.1. Setting alls=2 specifies
that this function should compute a set of gamma.group-orbit representatives of the required complete
subgraphs.

The optional parameter alls controls how many complete subgraphs are returned. The valid values for alls
are 0, 1 (the default), and 2.

If alls=0 (or false for backward compatibility) then K will contain at most one element. If maxi=false
then K will contain one element if and only if gamma contains a complete subgraph of size k . If maxi=true
then K will contain one element if and only if gamma contains a maximal complete subgraph of size k (in
which case K will contain (the vertex-set of) such a maximal complete subgraph).

If alls=1 (or true for backward compatibility) and maxi=false, then K will contain (perhaps properly) a
set of gamma.group orbit-representatives of the size k complete subgraphs of gamma. If alls=1 (the default)
and maxi=true, then K will contain (perhaps properly) a set of gamma.group orbit-representatives of the
size k maximal complete subgraphs of gamma.

If alls=2 and maxi=false, then K will be a set of gamma.group orbit-representatives of the size k complete
subgraphs of gamma. If alls=2 and maxi=true then K will be a set of gamma.group orbit-representatives
of the size k maximal complete subgraphs of gamma. This option can be more costly than when alls=1.

The optional parameter maxi controls whether only maximal complete subgraphs of size k are returned.
The default is false, which means that non-maximal as well as maximal complete subgraphs of size k are



Section 3. CompleteSubgraphsOfGivenSize 37

returned. If maxi=true then only maximal complete subgraphs of size k are returned. (Previous to version
4.1 of GRAPE, maxi=true meant that it was assumed (but not checked) that all complete subgraphs of size
k were maximal.)

The optional boolean parameter col is used to determine whether or not partial proper vertex-colouring is
used to cut down the search tree. The default is true, which says to use this partial colouring (and which
seems to be a good idea). For backward compatibility, col a rational number means the same as col=true.

If the optional parameter wts is given, then it is assumed to be a gamma.group invariant list (where the
action permutes the list positions) of length gamma.order of positive integer weights corresponding to the
vertices, in which case, a complete subgraph of size k means a complete subgraph whose vertex-weights sum
to k .

An alternative name for this function is CliquesOfGivenSize.

See also 7.2.1.

gap> gamma:=JohnsonGraph(6,2);
rec( isGraph := true, order := 15,
group := Group([ ( 1, 6,10,13,15, 5)( 2, 7,11,14, 4, 9)( 3, 8,12),

( 2, 6)( 3, 7)( 4, 8)( 5, 9) ]),
schreierVector := [ -1, 2, 2, 1, 1, 1, 1, 1, 2, 1, 1, 1, 1, 1, 1 ],
adjacencies := [ [ 2, 3, 4, 5, 6, 7, 8, 9 ] ], representatives := [ 1 ],
names := [ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 1, 5 ], [ 1, 6 ], [ 2, 3 ],

[ 2, 4 ], [ 2, 5 ], [ 2, 6 ], [ 3, 4 ], [ 3, 5 ], [ 3, 6 ], [ 4, 5 ],
[ 4, 6 ], [ 5, 6 ] ], isSimple := true )

gap> CompleteSubgraphsOfGivenSize(gamma,4);
[ [ 1, 2, 3, 4 ] ]
gap> CompleteSubgraphsOfGivenSize(gamma,4,1,true);
[ ]
gap> CompleteSubgraphsOfGivenSize(gamma,5,2,true);
[ [ 1, 2, 3, 4, 5 ] ]
gap> delta:=NewGroupGraph(Group(()),gamma);;
gap> CompleteSubgraphsOfGivenSize(delta,5,2,true);
[ [ 1, 2, 3, 4, 5 ], [ 1, 6, 7, 8, 9 ], [ 2, 6, 10, 11, 12 ],
[ 3, 7, 10, 13, 14 ], [ 4, 8, 11, 13, 15 ], [ 5, 9, 12, 14, 15 ] ]

gap> CompleteSubgraphsOfGivenSize(delta,5,0);
[ [ 1, 2, 3, 4, 5 ] ]
gap> CompleteSubgraphsOfGivenSize(delta,5,1,false,true,
> > [1,2,3,4,5,6,7,8,7,6,5,4,3,2,1]);
[ [ 1, 4 ], [ 2, 3 ], [ 3, 14 ], [ 4, 15 ], [ 5 ], [ 11 ], [ 12, 15 ],
[ 13, 14 ] ]

Automorphism groups
and isomorphism

testing for graphs

GRAPE provides a basic interface to B.D. McKay’s nauty (Version 2.0b5) package for calculating automor-
phism groups of (possibly vertex-coloured) graphs and for testing graph isomorphism (see [McK90]). To use
functions depending on nauty, GRAPE must be fully installed on a computer running UNIX (see 1.1).

8.1 AutGroupGraph

1I AutGroupGraph( gamma )
I AutGroupGraph( gamma, colourclasses )

The first version of this function returns the automorphism group of the (directed) graph gamma, using
nauty (this can also be accomplished by typing AutomorphismGroup(gamma)). The automorphism group
Aut (gamma) of gamma is the group consisting of the permutations of the vertices of gamma which preserve
the edge-set of gamma.

In the second version, colourclasses is an ordered partition of the vertices of gamma (into colour-classes),
and the subgroup of Aut (gamma) preserving this ordered partition is returned. The ordered partition should
be given as a list of sets, although the last set in the list may be omitted. Note that we do not require that
adjacent vertices be in different colour-classes.

gap> gamma := JohnsonGraph(4,2);
rec( isGraph := true, order := 6,

group := Group([ (1,4,6,3)(2,5), (2,4)(3,5) ]),
schreierVector := [ -1, 2, 1, 1, 1, 1 ], adjacencies := [ [ 2, 3, 4, 5 ] ],
representatives := [ 1 ],
names := [ [ 1, 2 ], [ 1, 3 ], [ 1, 4 ], [ 2, 3 ], [ 2, 4 ], [ 3, 4 ] ],
isSimple := true )

gap> Size(AutGroupGraph(gamma));
48
gap> Size(AutGroupGraph(gamma,[[1,2,3],[4,5,6]]));
6
gap> Size(AutGroupGraph(gamma,[[1,6]]));
16

8.2 IsIsomorphicGraph

1I IsIsomorphicGraph( gamma1, gamma2 )
I IsIsomorphicGraph( gamma1, gamma2, firstunbindcanon )

This boolean function uses the nauty package to test whether graphs gamma1 and gamma2 are isomorphic.
The value true is returned if and only if the graphs are isomorphic (as directed, uncoloured graphs).

The optional boolean parameter firstunbindcanon determines whether or not the canonicalLabelling com-
ponents of both gamma1 and gamma2 are first unbound before testing isomorphism. If firstunbindcanon is
true (the default, safe and possibly slower option) then these components are first unbound. If firstunbind-
canon is false, then any existing canonical labelling is used, which was the behaviour in versions of GRAPE
before 4.0. However, since canonical labellings can depend on the version of nauty (currently 2.0b5), certain
parameters of nauty (always set the same for a given graph by GRAPE 4.1), and the compiler and computer
used, you must be sure that if firstunbindcanon=false then the canonicalLabelling component(s) which
may already exist for gamma1 or gamma2 were created in exactly the same environment in which you are
presently computing. We remark that GRAPE 4.1 now sets nauty parameters differently than before for non-
simple graphs (and so canonical labellings may be different than before), in order to avoid a performance
problem with certain sparse directed graphs.

gap> gamma := JohnsonGraph(7,4);;
gap> delta := JohnsonGraph(7,3);;
gap> IsIsomorphicGraph( gamma, delta );
true



9 Partial Linear Spaces

Let s and t be positive integers. A partial linear space (P ,L), with parameters (s, t) consists of a set P
of points, together with a set L of (s + 1)-subsets of P called lines, such that every point is in exactly t + 1
lines, and every pair of distinct points is contained in at most one line. The point graph of a partial linear
space S having point-set P is the graph with vertex-set P and having [p, q ] an edge if and only if p 6= q and
p, q are in a common line of S . Two partial linear spaces (P ,L) and (P ′,L′) (with parameters (s, t)) are said
to be isomorphic if there is a bijection P → P ′ which induces a bijection L→ L′. An automorphism of
a partial linear space is an isomorphism onto itself. The set of all automorphisms of a partial linear space S
forms a group, called the automorphism group of S .

GRAPE contains a function PartialLinearSpaces to determine and classify partial linear spaces with given
point graph and parameters. In this chapter we describe this function, and also give a research application
of this function.

9.1 PartialLinearSpaces

1I PartialLinearSpaces( ptgraph, s, t )
I PartialLinearSpaces( ptgraph, s, t, nspaces )
I PartialLinearSpaces( ptgraph, s, t, nspaces, printlevel )
I PartialLinearSpaces( ptgraph, s, t, nspaces, printlevel, cliques )

This function classifies the partial linear spaces with given point graph ptgraph, and parameters (s,t). It
calls functions making use of the nauty package within GRAPE, and so can only be used on UNIX systems
on which GRAPE has been fully installed.

The function PartialLinearSpaces returns a list of representatives of distinct isomorphism classes of partial
linear spaces with (simple) point graph ptgraph, and parameters (s,t). The default is that representatives
for all isomorphism classes are returned.

The integer argument nspaces is optional, and has default value -1, which means that representatives for
all isomorphism classes are returned. If nspaces is non-negative then exactly nspaces representatives are
returned if there are at least nspaces isomorphism classes, otherwise representatives for all isomorphism
classes are returned.

In the output of this function, a partial linear space S is given by its incidence graph delta. The point-vertices
of delta are 1,...,ptgraph.order, with the name of point-vertex i being the name of vertex i of ptgraph. A
line-vertex of delta is named by a list (not necessarily ordered) of the point-vertex names for the points
on that line. We warn that this is a different naming convention to versions of GRAPE before 4.1. The
group delta.group associated with the incidence graph delta is the automorphism group of S acting on
point-vertices and line-vertices, and preserving both sets.

If printlevel is bound then it controls the print-level (default 0). Permitted values for printlevel are 0,1,2.

If cliques is bound then it is assumed to be a list (without repeats) of the (s + 1)-cliques of ptgraph. If
known, this can help the function to run faster.

gap> K7:=CompleteGraph(SymmetricGroup(7));;
gap> P:=PartialLinearSpaces(K7,2,2);
[ rec( isGraph := true, order := 14,

group := Group([ ( 1, 2)( 5, 6)( 9,11)(10,12),
( 1, 2, 3)( 5, 6, 7)( 9,11,13)(10,12,14),
( 1, 2, 3)( 4, 7, 6)( 9,12,14)(10,11,13),
( 1, 4, 7, 6, 2, 5, 3)( 8, 9,13,10,11,12,14) ]),

schreierVector := [ -1, 1, 2, 4, 4, 1, 3, -2, 4, 1, 1, 3, 4, 2 ],
adjacencies := [ [ 8, 9, 10 ], [ 1, 2, 3 ] ],
representatives := [ 1, 8 ],
names := [ 1, 2, 3, 4, 5, 6, 7, [ 1, 2, 3 ], [ 1, 4, 5 ], [ 1, 6, 7 ],

[ 2, 4, 6 ], [ 2, 5, 7 ], [ 3, 4, 7 ], [ 3, 5, 6 ] ],
isSimple := true ) ]

gap> Size(P[1].group);
168
gap> T:=ComplementGraph(JohnsonGraph(10,2));;
gap> P:=PartialLinearSpaces(T,4,6);;
gap> List(P,x->Size(x.group));
[ 216, 1512 ]

9.2 A research application of PartialLinearSpaces

We now provide an extended example of the use of GRAPE which illustrates a research application of the
PartialLinearSpaces function.

First we give a definition. Let s and t be positive integers. A partial geometry is a partial linear space
with parameters (s, t) for which there is an additional constant constant α > 0, such that, for every line l
and every point p not on l , there are exactly α lines through p meeting l in some point.

Our example shows that the Haemers partial geometry [Hae81] is uniquely determined (up to isomorphism)
by its point graph, as is the dual of the Haemers geometry (where the role of points and lines are inter-
changed), and that each of these geoemetries has automorphism group isomorphic to A7.

We first construct and study the Hoffman-Singleton graph, using the construction of Peter Cameron con-
tained in [Cam99]. We then construct the point graph of the Haemers partial geometry [Hae81] (this partial
geometry has (s, t) = (4, 17) and α = 2). The vertices of this point graph are the edges of the Hoffman-
Singleton graph, and two such vertices are adjacent in the point graph precisely when they are at distance
2 in the edge-graph of the Hoffman-Singleton graph (see [Hae81]). We then construct and classify (up to
isomorphism) all partial linear spaces with parameters (4, 17) having point graph isomorphic to that of the
Haemers partial geometry. We find that the Haemers partial geometry is the only possibility. It follows
from basic theory of partial geometries that the Haemers partial geometry is uniquely determined up to
isomorphism (as a partial geometry) by its point graph. We also show that the dual of the Haemers partial
geometry is also uniquely determined by its point graph. Thus far, the only proof of these results is by
GRAPE. Our example also shows that the Haemers partial geometry and its dual each has automorphism
group isomorphic to A7.

The total runtime (not including calls of nauty) was about 504 CPU-seconds on a Pentium II running at
350 MHz.

gap> RequirePackage("grape");

Loading GRAPE 4.1 (GRaph Algorithms using PErmutation groups),
by L.H.Soicher@qmul.ac.uk.

true
gap>
gap> OnSetsRecursive:=function(x,g)
> if not IsList(x) then
> return x^g;
> else
> return Set(List(x, y->OnSetsRecursive(y,g)));
> fi;
> end;;
gap>
gap> HofSingAdjacency := function(x,y)
> #
> # This boolean function returns true iff x and y are
> # adjacent in the Hoffman-Singleton graph, in Peter Cameron’s
> # construction.
> #
> if Size(x)=3 then # x is a 3-set
> if Size(y)=3 then # y is a 3-set
> return Intersection(x,y)=[]; # join iff disjoint
> else # y is a projective plane
> return x in y; # join iff x is a line of y
> fi;
> else # x is a projective plane
> if Size(y)=3 then # y is a 3-set
> return y in x; # join iff y is a line of x
> else # y is a projective plane
> return false; # don’t join
> fi;
> fi;
> end;;
gap>
gap> projectiveplane:=
> Set([[1,2,4],[2,3,5],[3,4,6],[4,5,7],[1,5,6],[2,6,7],[1,3,7]]);;
gap>
gap> HofSingGraph:=Graph(AlternatingGroup(7),
> [[1,2,3], projectiveplane], OnSetsRecursive,
> HofSingAdjacency);;
gap> GlobalParameters(HofSingGraph);
[ [ 0, 0, 7 ], [ 1, 0, 6 ], [ 1, 6, 0 ] ]
gap> autgrp := AutGroupGraph(HofSingGraph);;
gap> Size(autgrp);
252000
gap> HofSingGraph := NewGroupGraph(autgrp,HofSingGraph);;
gap> pointgraph:=DistanceGraph( EdgeGraph(HofSingGraph), 2);;
gap> GlobalParameters(pointgraph);
[ [ 0, 0, 72 ], [ 1, 20, 51 ], [ 36, 36, 0 ] ]
gap> spaces:=PartialLinearSpaces(pointgraph,4,17);;
gap> Length(spaces);
1
gap> haemers:=spaces[1];;
gap> DisplayCompositionSeries(haemers.group);
G (3 gens, size 2520)

A(7)
1 (0 gens, size 1)
gap> linegraph:=PointGraph(haemers, Adjacency(haemers,1)[1]);;
gap> spaces:=PartialLinearSpaces(linegraph,17,4);;
gap> Length(spaces);
1
gap> dualhaemers:=spaces[1];;
gap> DisplayCompositionSeries(dualhaemers.group);
G (4 gens, size 2520)
A(7)

1 (0 gens, size 1)
gap> quit;

